EGB240 Electronic Design

Assessment 1: PCB Alarm Circuit Design Portfolio

Submitted 3/04/2025

Adam Telfer (n12068373)

Executive Summary

This portfolio documents the design of a printed circuit board (PCB) for driving a piezoelectric buzzer circuit. The circuit is designed to meet the following specifications:

- Outputs a 2 kHz or 5 kHz tone, switching between them
- Switching frequency is variable at 2 Hz or 4 Hz, changing every 4 seconds
- Powered by two AA batteries (3V nominal)
- Slide Switch to toggle circuit on and off
- Physical dimensions: 55 x 40 x 12 mm (excluding battery holder and connector)

Contents

1. Circuit schematic



Figure 1: PCB circuit schematic. PCB layout (front and back) is shown in Figure 3.

2. Summary of design and operation

The circuit is designed to generate a 2 kHz or 5 kHz tone from a piezoelectric buzzer, with a switching frequency (the rate that it switches between the two tones) set at either 2 Hz or 4 Hz, with the switching frequency changing every 4 seconds. This is achieved using Schmitt trigger inverters to form relaxation oscillators; two tone oscillators, two switching oscillators and one to control which switching oscillator is active, thus five in total. The circuit is powered by a 2xAA battery pack, providing a nominal 3V. The pack has flying leads crimped to a 2-pin JST wire-to-board connector (shown in Figure 5), allowing for quick connection and disconnection of the bulkiest component.

A Schmitt trigger is a logic gate that has two threshold voltages; when the input is below the low threshold voltage the output will be low, and when above the high threshold voltage, the output will be high, with the difference between thresholds called hysteresis. The 74HC14 IC contains six Schmitt triggers with logic inverters, so when the input is above the high threshold the output is low, and when below the low threshold the output is high. An oscillator can be made by connecting a resistor between the input and output of an inverting Schmitt trigger, with a capacitor from input to ground. When the input is low, the output is high, charging the capacitor through the resistor until the high threshold is reached. The output then switches low, discharging the capacitor until the low threshold is reached, restarting the cycle. The oscillation period depends on the RC time constant and a K-factor, which can be found in the IC's datasheet or through testing, "However, its frequency is not particularly well determined, because hysteresis is not a well-controlled parameter in logic ICs"[1].

Switching Control (SC) oscillates at 0.25 Hz (4 second period) to allow differentiation between switching frequencies. It controls the switching oscillators by holding their inputs high, keeping them inactive (not oscillating) with low outputs. Switching A (SA) is active when SC is high and Switching B (SB) when SC is low. An additional gate inverts the SC signal to SA instead of a reverse biased diode, ensuring both switching oscillators have low outputs when inactive. If an inactive output were high, it would block signals from the active oscillator.

Switching A oscillates at 2 Hz, allowing at least two full cycles during its active time. When first becoming active, the capacitor must discharge from the IC's input voltage (minus the diode's forward voltage), during which Tone A (TA) is output. A switching frequency too close to SC causes a large proportion of TA output compared to TB which is undesirable, so two full cycles help minimize this effect (SB experiences this too but less noticeably due to its higher frequency).

Switching B oscillates at 4 Hz, double SA's frequency, chosen to be double SA, though it can be higher. The upper limit would be when SB's period becomes too short, and the two tones become indistinguishable. The outputs of SA and SB go to Tone Control (TC) which determines which tone oscillator is active.

[1] P. Horowitz and W. Hill, The Art of Electronics, 3rd ed. Cambridge, U.K.: Cambridge Univ. Press, 2015, pp. 237, 426, 427.

Tone A is active when TC is low, as its input diode (D4) is not forward biased, allowing normal oscillation. When TC is high, TA's input is high, keeping it inactive (like the switching oscillators).

Tone B (TB) will be inactive when TC is low, as current flows from TB's output, through R5, D6, R6 to ground preventing the capacitor from charging above the high threshold voltage. R5>>R6 so that most of the voltage drop occurs over R5, keeping TB's input below the high threshold, and TC below D4's forward voltage. R6 was set at 1 kΩ to balance minimising current draw from the outputs of SA and SB (2 mA in simulation), while not forcing R5 to be too high. When TC is high, D6 is reverse biased, stopping current flow and allowing C5 to charge and thus TB to oscillate. TB's output will be high when inactive, but this does not present an issue.

The frequency for the tone oscillators was chosen at 2 kHz and 5 kHz, as the square wave drive graph in Figure 2[2] showing spikes in the sound pressure at those frequencies, with a large difference between the frequencies to make them audibly distinct, even when accounting for a large difference between the design and real frequencies. Due to the need for R5>>R6, it is best to have the lower frequency tone as TB. The tone oscillators were able to acceptably drive the buzzer, so no additional circuity was needed to achieve a desirable volume from the buzzer.



Figure 2: Graphs of the Sound pressure vs Frequency of the PS1720P02 piezoelectric buzzer used in the design, with the square wave drive graph being of relevance due to Schmitt triggers outputting square waves [Ref. to data sheet].

The selection of resistance and capacitance was based on different factors for each oscillator type. For the SC oscillator, an electrolytic capacitor was chosen to accommodate larger capacitance values, making it easier to achieve lower frequencies if desired. The capacitor's higher tolerance was not a significant concern due to the low-frequency nature of this oscillator. A 3.3 uF capacitor was selected as a balance between minimizing size and cost while avoiding the need for an excessively high resistance (several MΩ). For the switching oscillators, a ceramic disk capacitor was used due to its tighter tolerance compared to electrolytics. A 1 uF capacitor was selected, as it is a common upper limit for ceramic disk capacitors while keeping the resistance under a MΩ for the required frequencies. For the tone oscillators, the resistor and capacitor values were constrained by two factors. The first is R5 needed to be greater than 10 kΩ (to ensure R5>>R6), and second was the capacitance had to remain above 10 nF to prevent excessive noise and to ensure that parasitic capacitance from PCB traces and other components did not significantly affect the oscillator's total capacitance. The decoupling capacitor (C6) was set to be 10 times the sum of all other capacitors (47 uF was the closest available until 100 uF).

[2] TDK Corporation. "Piezoelectronic Buzzer Catalog." TDK.com. Accessed: Apr. 2, 2025. [Online]. Available: https://product.tdk.com/system/files/dam/doc/product/sw_piezo/sw_piezo/piezo-buzzer/catalog/piezoelectronic_buzzer_ps_en.pdf.

3. PCB layout



Figure 3: PCB front (left) and back (right) layout. Front is component side; back is copper side. Overall dimensions shown in millimetres.

4. Bill of materials

Reference	Value	Description	Exclude from Board	Footprint		Qty
BT1	~	Battery holder, 3V, 2xAA, flying leads	Excluded from board	BATT-3V		1
C1	3.3u	Aluminium Electrolytic Capacitors - Radial Leaded 3.3uF 50V		CAP-RB-P2.54-D6.3		1
C2,C3	1u	Multilayer Ceramic Capacitors MLCC - Leaded 1uF 50V 10% 5mm LS		CAP-DISC-P5.08		2
C4,C5	10n	Multilayer Ceramic Capacitors MLCC - Leaded .01uF 50V 10% 2.54mmLS		CAP-DISC-P5.08		2
C6	47u	Aluminium Electrolytic Capacitors - Radial Leaded 100uF 25V		CAP-RB-P2.54-D6.3		1
D1,D2,D3,D4,D5,D6	1N4148	Standard switching diode, DO-35, 100V 0.15A		DO35-P7.62		6
J1	~	Wire-to-board connector, 01x02 Pin(Male), P-2.5mm, Solder Pin		JST B2B-XH-A		1
J2	~	Wire-to-board connector, 01x02 Pin(Female), P-2.5mm, Wire Crimp	Excluded from board	JST XHP-2		1
X1	~	Piezoelectric buzzer		PS1720P02		1
R1	1.5Meg	Metal Film Resistors - Through Hole 1.5M OHM 1/4W 1%		AXIAL-P10.16		1
R2	560k	Metal Film Resistors - Through Hole 560K OHM 1/4W 1%		AXIAL-P10.16		1
R3	270k	Metal Film Resistors - Through Hole 270K OHM 1/4W 1%		AXIAL-P10.16		1
R4	22k	Metal Film Resistors - Through Hole 56K OHM 1/4W 1%		AXIAL-P10.16		1
R5	56k	Metal Film Resistors - Through Hole 22K OHM 1/4W 1%		AXIAL-P10.16		1
R6	1k	Metal Film Resistors - Through Hole 1/4W 1K Ohm 1%		AXIAL-P10.16		1
S1	~	Switch, SPDT, Slide, On-On		SS-12		1
U1	74HC14	IC, Hex inverter schmitt trigger, DIP-14		DIP-14		1
U1	~	IC socket, DIP-14		DIP-14		1

Reference	Manufacturer	MPN	Supplier	SKU	MOQ	Price ($)
BT1	Generic	Battery holder, 3V, 2xAA, flying leads	Mouser	12BH222-GR	1	2.06
C1	Generic	Aluminium Electrolytic Capacitors - Radial Leaded 3.3uF 50V	Mouser	667-ECE-A1HKA3R3	1	0.32
C2,C3	Generic	Multilayer Ceramic Capacitors MLCC - Leaded 1uF 50V 10% 5mm LS	Mouser	594-K105K20X7RF5TH5	1	0.976
C4,C5	Generic	Multilayer Ceramic Capacitors MLCC - Leaded .01uF 50V 10% 2.54mmLS	Mouser	594-K103K15X7RF5TK2	1	0.256
C6	Generic	Aluminium Electrolytic Capacitors - Radial Leaded 100uF 25V	Mouser	667-ECA-1EHG101	1	0.496
D1,D2,D3,D4,D5,D6	Onsemi	1N4148	Mouser	512-1N4148	1	0.16
J1	JST Commercial	B2B-XH-A(LF)(SN)	Mouser	306-B2BXHALFSNP	1	0.16
J2	JST Commercial	XHP-2	Mouser	306-XHP-1P	1	0.37
X1	TDK	PS1720P02	Mouser	810-PS1720P02	1	1.15
R1	Generic	Metal Film Resistors - Through Hole 1.5M OHM 1/4W 1%	Mouser	603-MFR-25FRF52-1M5	1	0.16
R2	Generic	Metal Film Resistors - Through Hole 560K OHM 1/4W 1%	Mouser	603-MFR-25FBF52-560K	1	0.16
R3	Generic	Metal Film Resistors - Through Hole 270K OHM 1/4W 1%	Mouser	603-MFR-25FBF52-270K	1	0.16
R4	Generic	Metal Film Resistors - Through Hole 56K OHM 1/4W 1%	Mouser	603-MFR-25FBF52-56K	1	0.16
R5	Generic	Metal Film Resistors - Through Hole 22K OHM 1/4W 1%	Mouser	603-MFR-25FRF52-22K	1	0.16
R6	Generic	Metal Film Resistors - Through Hole 1K OHM 1/4W 1%	Mouser	603-MFR-25FTF52-1K	1	0.16
S1	NKK Switches	SS12SDP4	Mouser	633-SS12SDP4	1	4.5
U1	Texas Instruments	CD74HC14E	Mouser	595-CD74HC14E	1	1.02
U1	Generic	IC & Component Sockets, 14P DIP	Mouser	571-1-2199298-3	1	0.368

Table 1: Bill of Materials for PCB design.

5. Assembly overlay



Figure 4: Assembly overlay for PCB circuit design.

6. Photos of assembled prototype



Figure 5: Photograph of assembled prototype. Shown scale is in millimetres. Red is 3V, grey is GND. SC, SA, SB, TA and TB labels point to respective capacitors, TC is entire column connected by short brown wire.

7. Simulation circuit and results



External piezo buzzer model used
.include PS1720P02.sub
Due to long initial charge time, start SC_IN just below high threshold voltage
.ic V(SC_IN)=1.6
Schmitt Inverters modelled from 74HC14 specifications with 3V supply

F_t: Target Frequency
C: Capacitor to keep R 10k-1Meg (Minimum due to R5>>R6)
R_c: Calculated Resistance = 1/(K*F_t*C)
F_c: Calculated Frequency with actual resistor value = 1/(K*R*C)
R: Used component value close to R_c, single resistor used since exact frequency not critical
K-Factor = 0.9, derived from 74HC14 data sheet or testing of IC

Simulation 1: Switching Control, Switching A and B
.tran 0 4.75 0.05 uic

Simulation 2: Transition from Tone A to Tone B
.tran 0 443m 440m uic

Simulation 3: Transition from Tone B to Tone A
.tran 0 182m 179m uic

Figure 6: LTSpice simulation circuit.

Simulation Model Netlist (LTSpice)

```
* C:\Users\adamt\OneDrive\Documents\LTSpice\EGB240\Buzzer Circuit V10.asc
A1 SC_IN 0 0 0 0 Switching_Control 0 0 SCHMITT Vhigh=3 Rhigh=160 Rlow=80 Cout=200p Vt=1.325 Vh=0.325 td=31n
C1 SC_IN 0 3.3µ
R1 Switching_Control SC_IN 1.5Meg
A2 SA_IN 0 0 0 0 SA_OUT 0 0 SCHMITT Vhigh=3 Rhigh=160 Rlow=80 Cout=200p Vt=1.325 Vh=0.325 td=31n
C2 SA_IN 0 1µ
R2 SA_IN SA_OUT 560k
A3 SB_IN 0 0 0 0 SB_OUT 0 0 SCHMITT Vhigh=3 Rhigh=160 Rlow=80 Cout=200p Vt=1.325 Vh=0.325 td=31n
C3 SB_IN 0 1µ
R3 SB_OUT SB_IN 270k
A4 TA_IN 0 0 0 0 TA_OUT 0 0 SCHMITT Vhigh=3 Rhigh=160 Rlow=80 Cout=200p Vt=1.325 Vh=0.325 td=31n
C4 TA_IN 0 10n
R4 TA_OUT TA_IN 22k
A5 TB_IN 0 0 0 0 TB_OUT 0 0 SCHMITT Vhigh=3 Rhigh=160 Rlow=80 Cout=200p Vt=1.325 Vh=0.325 td=31n
C5 TB_IN 0 10n
R5 TB_OUT TB_IN 56k
D2 Switching_Control SB_IN 1N4148
A6 Switching_Control 0 0 0 0 N001 0 0 SCHMITT Vhigh=3 Rhigh=160 Rlow=80 Cout=200p Vt=1.325 Vh=0.325 td=31n
D1 N001 SA_IN 1N4148
D4 Tone_Control TA_IN 1N4148
D5 SB_OUT Tone_Control 1N4148
D3 SA_OUT Tone_Control 1N4148
D6 TB_IN Tone_Control 1N4148
R6 Tone_Control 0 1k
XX1 TA_OUT TB_OUT PS1720P02
.model D D
.lib C:\Users\adamt\AppData\Local\LTspice\lib\cmp\standard.dio
.ic V(SC_IN)=1.6
* F_t: 0.25\nC: 3.3u\nR_c = 1.347Meg\nF_c = 0.224
```

```
* F_t: 2\nC: 1u\nR_c = 556K\nF_c = 1.98
* F_t: 4\nC: 1u\nR_c = 278K\nF_c = 4.11
* F_t: 5k\nC: 10n\nR_c = 22.2K\nF_c = 5.05k
* F_s: 2k\nC: 10n\nR_c = 55.6K\nF_c = 1.98k
.include PS1720P02.sub
* Tone A
* Tone B
* Switching A
* Switching B
* Switching Control
* Tone Control
.tran 0 4.75 0.05 uic
* External piezo buzzer model used
* .tran 0 182m 179m uic
* Simulation 1: Switching Control, Switching A and B
* Simulation 2: Transition from Tone A to Tone B
* Buzzer
* Simulation 3: Transition from Tone B to Tone A
* .tran 0 443m 440m uic
* K-Factor = 0.9, derived from 74HC14 data sheet or testing of IC
* F_c: Calculated Frequency with actual resistor value = 1/(K*R*C)
* R_c: Calculated Resistance = 1/(K*F_t*C)
* C: Capacitor to keep R 10k-1Meg (Minimum due to R5>>R6)
* F_t: Target Frequency
* Schmitt Inverters modelled from 74HC14 specifications with 3V supply
* Due to long initial charge time, start SC_IN just below high threshold voltage
* R: Used component value close to R_c, single resistor used since exact frequency not critical
.backanno
.end
```

Simulation Results



Figure 7: LTSpice simulation showing that when Switching Control (top, magenta) is high Switching A (bottom, blue) is active with a frequency of 1.80 Hz.



Figure 8: LTSpice simulation showing that when Switching Control (top, magenta) is low Switching B (bottom, blue) is active with a frequency of 3.70 Hz.



Figure 9: LTSpice simulation showing Switching Control (top, magenta) oscillates with a frequency of 0.225 Hz. Tone Control (bottom, blue) is the combined outputs from Switching A and Switching B.



Figure 10: LTSpice simulation showing the transition from Tone A to Tone B (bottom, blue, Tone B output minus Tone A output), controlled by Tone Control (top, magenta) going from low to high. Tone B has a frequency of 1.97 kHz.



Figure 11: LTSpice simulation showing the transition from Tone B to Tone A (bottom, blue, Tone B output minus Tone A output), controlled by Tone Control (top, magenta) going from high to low. Tone A has a frequency of 4.90 kHz.

8. Experimental results

Figure 12: Oscilloscope capture showing that when Switching Control (top, yellow) is high Switching A (bottom, green) is active with a frequency of 1.63 Hz.



Figure 13: Oscilloscope capture showing that when Switching Control (top, yellow) is low Switching B (bottom, green) is active with a frequency of 3.51 Hz.

Figure 14: Oscilloscope capture showing Switching Control (top, yellow) oscillates with a frequency of 0.192 Hz. Tone Control (bottom, green) is the combined outputs from Switching A and Switching B.



Figure 15: Oscilloscope capture showing the transition from Tone A to Tone B (bottom, blue, Tone B output minus Tone A output), controlled by Tone Control (top, yellow) going from low to high. Tone B has a frequency of 1.83 kHz.



Figure 16: Oscilloscope capture showing the transition from Tone B to Tone A (bottom, blue, Tone B output minus Tone A output), controlled by Tone Control (top, yellow) going from low to high. Tone A has a frequency of 4.44 kHz